

XALQARO NORDIK UNIVERSITETI

Iqtisodiyot va pedagogika fakulteti, Iqtisodiyot va biznesni boshqarish kafedrasi

Fan o‘qituvchisi: Sabirov Xasan Nusratovich

Mavzu: Dinamik qatorlar regressiyasi

Reja:

- 1. Asosiy tushunchalar**
- 2. Cheklangan taqsimlangan kechikishlar.**
- 3. Ketma-ket korrelyatsiya.**
- 4. Avtoregressiv taqsimlangan kechikish(lagli) modellari**
- 5. Multiplikator tahlili**

9.1. Asosiy tushunchalar

“Principle of Econometrics” 4-nashr adabiyotning 9-bobidagi kabi, dinamikaning regressiya munosabatlariga kirishi mumkin bo`lgan uchta usul ko`rib chiqamiz - tushuntirish o`zgaruvchining kechikish qiymatlari, bog`liq o`zgaruvchining kechikish qiymatlari va ulardagi xatoning kechikish qiymati.

Vaqtli qatorlarning regressiyasida odatiy ekonometrik jarayonlar tegishli statistik xususiyatlarga ega bo`lishi uchun ma`lumotlar statsionar bo`lishi kerak. Bu, asosan, vaqtli qatorlar ma`lumotlarining o`rtacha dispersiyasi va kovariatsiyalari ular kuzatilgan vaqt davriga bog`liq bo`lmasligi talab qiladi. Masalan, 1973 yilning uchinchi choragida YAIM ning o`rtacha va dispersiyasi 2006 yilning 4-choragidan farq qilishi mumkin emas. Ushbu muammoni hal qilish usullari so`nggi yillarda ekonometriklar uchun boy tadqiqot sohasini taqdim etdi va bu usullarning bir nechta 12- bobda keyinroq batafsil to`xtalib o`tamiz.

Qo’llaniladigan birinchi tahlil vositalaridan biri bu ma`lumotlarning oddiy vaqtli qatorlarni grafigidir. Vaqtli qatorlar grafigi ma`lumotlar bilan yuzaga kelishi mumkin bo`lgan muammolarni ochib beradi va statistik jihatdan davom etish yo’llarini taklif qiladi. Oldingi boblarda ko`rib chiqganimizdek, Stata-da vaqtli qatorlar grafiklarini yaratish osон va quyida bir nechta yangi grafiklar ko`rib chiqiladi.

Nihoyat, bu bobda vaqtli ketma-ket kuzatuvlardan bilan bog`liq bo`lgani sababli kuzatuvlarning odatiy soni, N , ko`proq qo'llaniladigan T bilan almashtiriladi. Keyingi boblarda ham vaqt- seriyasi, ham kesma ma'lumotlar qo'llaniladi, N va T dan foydalaniladi.

9.1.1. Stata-da vaqtli qatorlar grafigini ifodalash

Vaqtli qatorlar ma'lumotlarini tahlil qilish uchun Stata-ning ko'plab o'rnatilgan funktsiyalaridan foydalanish uchun, bir to'plamdagagi ma'lumotlarni vaqtli qatorlar deb e'lon qilishi kerak. Vaqtli qatorlarni o'z vaqtida tartiblanganligi sababli boshqa kuzatuvlarga nisbatan pozitsiyasi saqlanishi kerak. Axir bu turdagagi ma'lumotlarni tahlil qilish, kesma tahlilidan farq qiladigan narsa ularning vaqtinchalik munosabatlaridir.

Agar sizda mavjud bo'lgan ma'lumotlar kuzatuv to`plangan vaqtini aniqlash uchun tegishli sanaga ega bo`lmasa, uni qo'shish yaxshi fikrdir. Bu tarixni aniqlash imkonini beradi, davrlarni osonlashtiradi va grafiklarning axborot mazmunini sezilarli darajada oshiradi. Kitobda ifodalanilgan ma'lumotlar to`plamalari vaqtli qatorlar deb e'lon qilinmagan va ularning aksariyati o`zgaruvchilar to`plamida tegishli sanalarni o'z ichiga olmaydi. Shunday qilib, biznesning birinchi tartibi bu ma'lumotni ma'lumotlar to'plamiga qo'shishdir va keyin kuztuvlarni aniqlash uchun sanalardan foydalanishdir (masalan kunlik, oylik, choraklik, yillik). Ma'lumotlarning vaqtga bog`liqligini tahlil qilishda, bu juda muhim ma'lumot bo`lib, quyida tushintiriladi.

Matndan aniq misollarga o'tishdan oldin, Stata sana va vaqtlnarni qanday boshqarishi haqida qo'shimcha ma'lumot kiritish kerak. Butun son kelishilgan asosdan (ularni qanday belgilashingizdan qat'iy nazar) o'tgan vaqt birliklari sonini yozadi, Stata uchun bu 1960

Masalan, 1961 yilda boshlangan 100 ta choraklik ma'lumotlarni kuzatish uchun biz Stata sanalarini yaratishimiz mumkin:

set obs 100

generate date = tq(1961q1) + _n-1

tq(1961q1) psevdofunktsiya deb ataladi. Ular psevdofunksiyalar deb atalishi, ular yozganlaringizni butun son ekvivalentlariga tarjima qiladi. 1961q1 ning butun ekvivalenti 4 ga teng, ya'ni 1960 yil birinchi chorakdan necha chorak o'tdi. Ikkinchi chorak 5 va shunga o'xshash tarzda o'rnatiladi. **_n-1** qo'shilishi kuzatuvlarni 1 ga oshirish uchun amalga oshiriladi. Ro'yhatning birinchi 5ta kuzatuvi sanani ko'rib chiqamiz:

. list date in 1/5

date	
1.	4
2.	5
3.	6
4.	7
5.	8

bu biz kutkan narsadir. Buni odamlar uchun mazmunli qilish uchun, ularni amalga oshirish uchun satrlar sifatida formatlash kerak 1960 yildan 20 chorak qaysi sana ekanligini aytish biz uchun oson. Bu **format** buyrug'i yordamida amalga oshiriladi.

format %tq date

Format buyrug'i butun sanalarni ko'rsatish usulini o'zgartiradi.

```
. list date in 1/5
```

date	
1.	1961q1
2.	1961q2
3.	1961q3
4.	1961q4
5.	1962q1

Ko'rib turganingizdek, **format %tq date** Stata-ga 4, 5, 6 va 7-sonlarni ko'rsatishni buyuradi va o'zgaruvchan **date** 1961q1, 1961q2 va hokazo. Nihoyat, kuzatishlar vaqtli qatorlar deb e'lon qilinadi, vaqt o'zgaruvchisini identifikatsiya qiluvchi o'zgaruvchi nomidan keyin **tsset** buyrug'i yordamida amalga oshiriladi.

tsset date

```
. tsset date
      time variable: date, 1961q1 to 1985q4
                  delta: 1 quarter
```

Ma'lumotlar vaqtli qatorlar deb e'lon qilingandan so'ng, Stata qamrab olingan davr va o`lchov oraliq i haqidagi muhim ma'lumotlarni chop etadi. U vaqt o`zgaruvchisining nomini, u qamrab olgan sanalarni va **delta** yoki kuzatishlar orasidagi vaqtni aniqlaydi. Yaratilgan sanalar kerakli narsaga mos kelishiga ishonch hosil qilish uchun har doim sanalarni yaratishda ehtiyyotkorlik bilan buni tekshiring. Stata haftalik (**tw**), oylik (**tm**) yillik (**ty**) va boshqalar ni aniqlash uchun boshqa funktsiyalar va psevdofunktsiyalarni o'z ichiga oladi. Shunga qaramay, ular **1960q1** yildan buyon o`tkan davrlarini ko'rsatadigan butun sonlar to`plamini yaratadi. Butun sonlarni sana sifatida ko'rsatish uchun mos formatlar (**%tw**), (**%tm**) va (**%ty**) mos ravishda. Boshqa variantlarni ko'rish va ular turi qanday ishlashi haqida ko'proq ma'lumot olish uchun quyidagi buyruqdan foydalanamiz:

help dates and times

Command oynasida va Stata **viewer** oynasini ochadi hamda sizni tegishli ma'lumotlarga olib boradi. Sanalar yaratilgandan va ma'lumotlar to'plami vaqtli qatorlar deb e'lon qilingandan so'ng, ma'lumotlar to'plamini saqlang shuning uchun bu jarayon ushbu ma'lumotlar uchun takrorlanishi shart emas. Stata yangi o'zgaruvchini, kerakli display formati va ma'lumotlar to'plami bilan birga vaqtli qatorlar ma'lumotlarini saqlaydi.

save new.dta, replace

O'zgartirish birikmasi Stata bir xil nomdagi mavjud ma'lumotlar to'plamini qayta yozishga yordam beradi, shuning uchun ushbu parametr dan ehtiyyot bo`ling.

Okun ma'lumotlar to'plami

Qilish kerak bo`lgan birinchi narsa – katalogni ma`lumotlaringizni o`z ichiga olgan katalogga o`zgartirish va ma`lumotni yuklash. Shu mashqda biz *okun.dta* ma`lumotlaridan foydalanamiz.

use okun, clear

Ushbu ma`lumotlar to'plamida foiz bo'yicha choraklik kuzatuvlar bo`lgan **g** va **u** ikkita o`zgaruvchi mavjud, bular mos ravishda 1985q2- 2009q3 yillarda AQSHda Groos Domestik va ishsizlik darajasining har choraklik kuzatuvlaridir. Ma`lumotlar yuklangandan so`ng, yaratish buyrug'i yordamida sana tayinlanadi. Stata sanalarni yaratish uchun maxsus funktsiyalarni o`z ichiga oladi, ular Stata sanalarga qanday munosabatda bo`lishini (butun sonlar) va odamlarning ishslash usullari (kunlar, oylar, yillar va boshqalar) tarjima qiladi.

Har choraklik ma`lumotlar 1985 yilning ikkinchi choragida boshlanadi. Sanalarni belgilash va barcha o`zgaruvchilarni vaqtli qatorlardan foydalanishga aylantirish.

generate date = tq(1985q2) + _n-1

list date in 1

format %tq date

list date in 1

tsset date

Natija:

```
. generate date = tq(1985q2) + _n-1
. list date in 1

      date
1.    101

: format %tq date
. list date in 1

      date
1.  1985q2

: tsset date
      time variable: date, 1985q2 to 2009q3
      delta: 1 quarter
```

Stata nima qilayotganini ko'rsatish uchun ikkita **list in 1** buyruqlar qo`shiladi - ular amaliyotda kerak emas. Shunga qaramay, ular 1985q2 1960q1dan 101 chorak oldinda ekanligini aniqladilar. Format buyrug'i Stata-ga 101 sanasini 1985q2 sifatida ko'rsatishni ifodalandaydi.

9.1.2. Vaqtli qatorlar grafiklari

Ma`lumotlar yuklangandan so`ng, vaqt o`zgaruvchisi yaratilgan, formatlangan, va o`zgaruvchilar vaqtli qatorlar sifatida e`lon qilinganidan so`ng, siz tahlilning dastlabki bosqichlarini boshlashga tayyorsiz. Vaqt ketma- ketligi bilan, vaqtga nisbatan o`zgaruvchilarni chizishdan ko`ra boshlash uchun yaxshiroq joy yo`q. Bu ma`lumotlar muhim xususiyatlarni ochib beradi (masalan, statsionarlik, tendentsiyalar, tizimli tanaffuslar va boshqalar). Ishsizlik darajasi va YaIM ning o'sish sur'atlarini aniqlash uchun **tsline**

ishlatiladi. Ikkala uchastkaning teglarini bir xil grafikda olish uchun teglar ***label var*** buyruqlari yordamida qisqartiriladi. Keyin ***tsline*** ya`ni ***graph twoway tsline*** qisqartmasi bo`lib, ikkala qatorni bir xil grafikda ifodalanadi.


label var u "% Unemployed"

label var g "% GDP growth"

tsline u g, lpattern(solid dash)

Ikki vaqtli qatorlar grafigi ustiga qo`yiladi, chunki grafik chiziladigan qatorlarning har biri qavs ichiga olinadi. Boshqa variantlardan foydalanish mumkin, ammo biz buni hozircha soddalashtiramiz.

Stata grafiklari quyidagicha ifodalanadi:



Ishsizlik grafigi (qalin chiziq) YAIM o`sishiga qaraganda keng diapazondagi o`zgarishlarni ko`rsatadi, lekin bir vaqt oralig`idan ikkinchisiga kamroq farq qiladi. O`zgaruvchilarning har biri statsionar emasligini ko`rsatadigan aniq tendentsiyalar, tannaffuslar yoki boshqa xususiyatlar mavjud emas. Shuning uchun bu o`zgaruvchilar, ehtimol, ushbu bobda muhokama qilingan an`anaviy regressiya usullari uchun juda mos keladi. Bobda grafiklarning mumkin bo`lgan nostatsionarligini o`rganish uchun yana 12 ta rasmiy test ishlab chiqilgan. Hozirgacha ular statsionar ekanligi taxmin qilinadi.

9.1.3. Stataning lagli va farqli o`zgaruvchilari

Ilgari ko`rilganidek, list buyrug'i ma'lumotlar to'plamidan ekranga o`zgaruvchilarni chop etish uchun foydalilanadi. Bunda kuzatuvlarni cheklash uchun ***in 1/5*** va ***96/98*** oralig'ida qo'llaniladi. Chop etilgan o`zgaruvchilar birinchi bo`lib 5-bobda o`rganilgan Stata unary operatorlarining boshqa namunasidan foydalilanadi.

Stata maxsus unary operatorlarini o`z ichiga oladi, ular vaqtli qatorlardagi ma'lumotlarning laglari va farqlarini juda oson hamda samarali qilish uchun foydalilanishi mumkin. Bu o`zgaruvchilar *Stata User's Manual* qo'llanmasida *Time-series varlists*

sarlavhasi ostida to`liq ifodalanilgan. Bu yerda operatorlarning qisman ro'yxati va ularning ma'nolari:

operator	Meaning
L.	lag (x_{t-1})
L2.	2-period lag (x_{t-2})
...	
D.	difference ($x_t - x_{t-1}$)
D2.	difference of difference ($x_t - 2x_{t-1} + x_{t-2}$)

Bu (unary) operatorlar davrdan keyingi o'zgaruvchisida ishlaydi. Masalan, $L.u$ o'zgaruvchini oladi u va uni bir davrga kechiktiradi(laglanadi). Xuddi shunday $D.u$ bir davr vaqt farqini $u_t - u_{t-1}$ oladi.

Kechikish(laglash) va farq operatorlari chiziqli bo'lib, har qanday tartibda birlgilikda ishlatilishi mumkin. Masalan, u (i.e., $ldu_t = u_{t-1} - u_{t-2}$) dagi kuzatuvlar orasidagi kechikkan farqni bir marta olish uchun $L.D.u$ dan foydalanish mumkin. Bu o'ngdan chapga ishlaydi: u farqini oling va keyin uni bir davrga kechiktiring(laglang). Operatsiyalarda chiziqlilik bu $D.L.u$ ekvivalentligini bildiradi - kechikish u bir davrda va keyin farq qiladi. Bu shuningdek to`g`ri $L.L = L2$. O'zgaruvchini ikki davrga kechiktirish uchun $L.L.u$ yoki soddaroq qilib aytganda, $L2.u$ dan foydalaning. Raqam quyidagi o'zgaruvchining o'tmishda qancha davr ortda qolganligini ko'rsatadi. Shunday qilib $L2.u$, u ikki davrga ortda qoladi (i.e., $= u_{t-2}$).

Yetakchi (F) va mavsumiy farqlar (S) ni yaratadigan qo'shimcha vaqtli qatorlar operatorlari mavjud. Faktor o'zgaruvchilari uchun unary operatorlar misolida bo`lgani kabi, bu vaqtli qator operatorlari ham modelga kiritish uchun alohida o'zgaruvchilar yaratish zaruriyatidan xalos qiladi. Quyida muhokama qilinadigan bir qancha boshqa yorliqlar mavjud.

Ushbu operatorlardan foydalanishni ko`rsatish uchun o'zgaruvchilar, kechikishlar va farqli ma`lumotlar to`plamining boshida hamda oxirida kuzatuvlar uchun quyida keltirilgan. Umuman olganda, ketma-ketlik mazmuni ma`noli bo`lishini va vaqt davrlari to`g`ri o'zgaruvchilarga tayinlanganligini ta`minlash uchun bir nechta kuzatuvlarni chop etish yaxshi amaliyotdir. Sana u ostida u, g , o'zgarishi va bir nechta kechikishlar vaqtli qatorlar operatorlari yordamida chop etiladi. Bular Principle of Econometrics 4-th (POE4) adabiyotida 9.1-jadvaldagagi kuzatuvlarga mos keladi.

list date u L.u D.u g L1.g L2.g L3.g in 1/5

list date u L.u D.u g L1.g L2.g L3.g in 96/98

. list date u L.u D.u g L1.g L2.g L3.g in 1/5

	date	u	L. u	D. u	L. g	L2. g	L3. g
1.	1985q2	7.3			1.4	.	.
2.	1985q3	7.2	7.3	- .1	2	1.4	.
3.	1985q4	7	7.2	- .2	1.4	2	1.4
4.	1986q1	7	7	0	1.5	1.4	2
5.	1986q2	7.2	7	.2	.9	1.5	1.4

. list date u L.u D.u g L1.g L2.g L3.g in 96/98

		L.	D.	L.	L2.	L3.
	date	u	u	g	g	g
96.	2009q1	8.1	6.9	1.2	-1.2	-1.4
97.	2009q2	9.3	8.1	1.2	-.2	-1.2
98.	2009q3	9.6	9.3	.3	.8	-.2
						-1.4

Vaqtli qatorlar o`zgaruvchilari ulardan foydalanishni osonlashtiradigan yana bir xususiyatga ega. Statada *operator(numlist)* buyrug`idir.

A *numlist* is a list of numbers with blanks or commas in between. There are a number of shorthand conventions to reduce the amount of typing necessary. For instance:

2	just one number
1 2 3	three numbers
3 2 1	three numbers in reversed order
5 1 1.5	three different numbers
1 3 -2.17 5.12	four numbers in jumbled order
1/3	three numbers: 1, 2, 3
3/1	the same three numbers in reverse order
5/8	four numbers: 5, 6, 7, 8
-8/-5	four numbers: -8, -7, -6, -5
-5/-8	four numbers: -5, -6, -7, -8
-1/2	four numbers: -1, 0, 1, 2
1 2 to 4	four numbers: 1, 2, 3, 4
4 3 to 1	four numbers: 4, 3, 2, 1
10 15 to 30	five numbers: 10, 15, 20, 25, 30
1 2:4	same as 1 2 to 4
4 3:1	same as 4 3 to 1
10 15:30	same as 10 15 to 30
1(1)3	three numbers: 1, 2, 3
1(2)9	five numbers: 1, 3, 5, 7, 9
1(2)10	the same five numbers: 1, 3, 5, 7, 9
9(-2)1	five numbers: 9, 7, 5, 3, and 1
-1(.5)2.5	the numbers: -1, -.5, 0, .5, 1, 1.5, 2, 2.5
1[1]3	same as 1(1)3
1[2]9	same as 1(2)9
1[2]10	same as 1(2)10
9[-2]1	same as 9(-2)1
-1[.5]2.5	same as -1(.5)2.5
1 2 3/5 8(2)12	eight numbers: 1, 2, 3, 4, 5, 8, 10, 12
1,2,3/5,8(2)12	the same eight numbers
1 2 3/5 8 10 to 12	the same eight numbers
1,2,3/5,8,10 to 12	the same eight numbers
1 2 3/5 8 10:12	the same eight numbers

Ko'rib turganingizdek, numlist juda moslashuvchan. U diapazonlarni, ketma-ketliklarni, shuningdek, aniq raqamlar ro'yxatini belgilash imkonini beradi. Bular manfiy raqamlarni o'z ichiga olishi mumkin va ularning tartibi osongina o`zgartirilishi mumkin. Ushbu sintaksidan foydalanib, list buyruqlarini qisqartirish mumkin.

list date L(0/1).u D.u L(0/3).g in 1/5

list date L(0/1).u D.u L(0/3).g in 96/98

L(0/1).u buyrug'i *u L.u* va *L(0/3).g* ga ekvivalent *g L.g L2.g L3.g* bilan bir xildir.

9.2. Cheklangan taqsimlangan kechikishlar(laglar)

Cheklangan taqsimlangan kechikish modellari mustaqil o`zgaruvchilarni va ularning kechikishlarini o`zgaruvchilar sifatida o'z ichiga oladi.

$$y_t = \alpha + \beta_0 x_t + \beta_1 x_{t-1} + \beta_2 x_{t-2} + \dots + \beta_q x_{t-q} + e_t, \quad t = q+1, \dots, T$$

Bu yerda ko'rib chiqilgan maxsus misol Okun qonunini tekshirishdir. Bu modelda ishsizlikning bir davrdan ikkinchi davrgacha o`zgarishi iqtisodiyotdagi ishlab chiqarishning o'sishi sur`atiga bog`liqligini ifodalaydi.

$$U_t - U_{t-1} = -\gamma(G_t - G_N)$$

Bu yerda U_t – ishsizlik darajasi, G_t – YaIM o'sishi, G_N – YaIM o'sishning normal sur'ati. Regressiya modeli:

$$DU_t = \alpha + \beta_0 G + e_t$$


Bu yerda D-farq operatori, $\alpha = \gamma G_N$ $\beta_0 = -\gamma$ va xato atamasi modelga qo'shilgan. Ishlab chiqarish hajmidagi o'zgarishlar ishsizlikka taqsimlangan kechikish ta'siriga ega bo`lishi mumkinligini tan olgan holda – barcha ta'sirlar bir zumda sodir bo`lmaydi – ishlab chiqarish uchun modelga kechikishlar(laglar) qo'shiladi.

$$DU_t = \alpha + \beta_0 G_t + \beta_1 G_{t-1} + \beta_2 G_{t-2} + \dots + \beta_q G_{t-q} + e_t, \quad t=q+1, \dots, T$$

Ikki vaqtli ketma-ketlikdan foydalananib grafik hosil qilish mumkin.

tsline D.u g

Quyidagi ko`rinishdagi grafik hosil bo`ladi.



Stata-da cheklangan taqsimlangan kechikish modelini hisoblash vaqtli qatorlar operatorlari yordamida juda oddiy amalga oshiriladi.

$q=3$ bo'lsin va

regress D.u L(0/3).g

Natija:

. regress D.u L(0/3).g					
Source	SS	df	MS	Number of obs = 95	
Model	5.13367789	4	1.28341947	F(4, 90) = 42.23	
Residual	2.73516422	90	.030390714	Prob > F = 0.0000	
Total	7.86884211	94	.083711086	R-squared = 0.6524	
				Adj R-squared = 0.6370	
				Root MSE = .17433	
D.u	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
g	-.2020526	.0330131	-6.12	0.000	-.267639 -.1364663
--.					
L1.	-.1645352	.0358175	-4.59	0.000	-.2356929 -.0933774
L2.	-.071556	.0353043	-2.03	0.046	-.1416941 -.0014179
L3.	.003303	.0362603	0.09	0.928	-.0687345 .0753405
_cons	.5809746	.0538893	10.78	0.000	.4739142 .688035

Yana bir bor **$L(numlist)$** sintaksisi **g** ning bir vaqtida va 3 ta kechikkan qiymatlarini joylashtirish uchun modelda ishlatiladi.

Ikki o'zgaruvchining kechikish qiymatidan foydalangan holda modelni qayta baholaymiz.

. regress D.u L(0/2).g

Source	SS	df	MS	Number of obs	=	96
Model	5.17925206	3	1.72641735	F(3, 92)	=	57.95
Residual	2.74074794	92	.029790739	Prob > F	=	0.0000
Total	7.92	95	.083368421	R-squared	=	0.6539
				Adj R-squared	=	0.6427
				Root MSE	=	.1726

D.u	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
g	-.2020216	.0323832	-6.24	0.000	-.2663374 -.1377059
--.					
L1.	-.1653269	.0335368	-4.93	0.000	-.2319339 -.0987198
L2.	-.0700135	.0331	-2.12	0.037	-.1357529 -.0042741
_cons	.5835561	.0472119	12.36	0.000	.4897892 .6773231

g dagi statistik jihatdan ahamiyatsiz uchunchi kechikishning pasayishi natijasida model mosligida deyarli hech qanday o`zgarish yo`q.

9.3. Ketma-ket korrelatsiya

Vaqtli qatorlar regressiyadagi yana bir murakkablik, regressiya modelining xatolari bir-biri bilan bog'liq bo'lganda yuzaga keladi. Bu Gauss -Markov teoremasining asosiy tahminlaridan birini buziladi va parametrlearning eng kichik kvadratlarini baholash xususiyatlariiga sezilarni ta'sir ko`rsatadi.


Iqtisodiyotda ketma-ket korrelyastsiya iqtisodiy tebranishlar davomiyligi ma`lumotlarning tanlab olish chastotasidan oshib ketganda sodir bo`ladi. Bu tebranish vaqt oralig`ida yo`qotishlarni olib keladi, bu esa xatolarning ijobiy bog`lanishiga olib keladi. Ko`p hollarda bu regressiyasning vaqt strukturasini to`g`ri modellashning muvaffaqiyatsizligini anglatadi - yoki kiritilgan o`zgaruvchilar bilan bog`liq bo`lgan kechikkan o`zgaruvchilar o`tkazib yuboriladi yoki to`g`ri modellashtirilmagan bog`liq o`zgaruvchida ba`zi bir barqarorlik mavjud bo`ladi. Yechim regressiya funktsiyasini to`g`ri belgilashdir $E(e_t | \text{barcha o`zgaruvchilar})=0$. Bu eng kichik kvadratlarning izchil bo'lishi kesishma va qiyaliklar uchun zarur shartni qondiradi.

Eng kichik kvadratlar qoldiqlarida avtokorrelyatsiyani aniqlash muhim, chunki eng kichik kvadratlar bu holatda nomuvofiq bo'lishi mumkin. Birinchi vosita **g** va **L.g** ning tarqalish grafigini ishlab chiqarish uchun ishlataladi. Gorizontal va vertikal chiziqlar taxminan o'rta joylashtiriladi.

summarize g

scatter g L.g, xline(`r(mean)') yline(`r(mean)')

Natija esa:



Grafikdan so`ng ***summarize*** buyrug'ini amalga oshirish zarur, chunki YaIM o'sishi o'rtacha grafikda ko'rsatilgan chiziqlarni chizish uchun hisoblash kerak. Saqlanganlar orasida ***g*** ning o'rtachasi natijalari, ularni odatdagagi ***return list*** buyrug'i yordamida ko'rish mumkin.

```
. summarize g
      variable |   obs      mean    std. dev.      min      max
      g        |   98  1.276531  .6469279  -1.4     2.5
. return list
scalars:
      r(N) = 98
      r(sum_w) = 98
      r(mean) = 1.276530612244898
      r(var) = .4185156743109615
      r(sd) = .6469278741180978
      r(min) = -1.4
      r(max) = 2.5
      r(sum) = 125.1
```


O'rtachaga kirish uchun uning katta nomi, ***r(mean)***, bittali qo`shtirnoq ichiga olinishi kerak, ya'ni, '***r(mean)***'. Birinchi qo`shtirnoq chap bitta qo`shtirnoq (`-- klaviyaturalarning yuqori chap tomonida), ikkinchisi esa o`ngdagi yagona qo`shtirnoq ('--located under the double quote " ko`pgina klaviyaturalarda ostida joylashgan).

Raqamlı yondashuv hisoblangan namunadagi avtokorrelyatsiyalarni ko'rib chiqishdir. Bular quyidagi buyruq orqali topiladi:

ac g, lags(12) generate(ac_g)

list ac_g in 1/12

ac buyrug`i keyingi (***g***) o`zgaruvchisi uchunchi namunaviy avtokorrelyatsiyalarni hisoblaydi va lags(12) birikmasi Stata-ga 12-davrgacha bo`lgan ***g*** oraliqda kompyuter avtokorrelyatsiyaslarini bildiradi. Natijalar grafikda ifodalaniladi, garchi avtokorrelyatsiyalar ***ac_g*** nomli o`zgaruvchidagi ***generate*** birikmasi yordamida saqlanadi.



Soyali maydonda 95 % ishonch bandi paydo bo`ladi. E'tibor bering, faqat birinchi ikkitasi avtokorrelyatsiyalar 5% darajasida noldan sezilarli darajada farq qiladi.

Taxminan 95% ishonch diapazonlari $\sqrt{Tr_k}^a \sim N(0,1)$ faktidan foydalanib hisoblanadi. **gen** dan **z=sqrt(e(N))*ac_g** chegara hosil qilish uchun foydalaning. Agar raqamlardan biri 1,96 dan kichik yoki 1,96 dan katta bo`lsa, u taxminan 95% ishonch oralig`idan tashqarida yotadi va 5% darajasida statistik ahamiyatga ega bo`ladi. Stataning ishslash funksiyasi boshqa (*Bartlett's*) metoddan foydalanadi va natijalar ushbu oddiy taxminga asoslanganidan farq qilishi mumkin.

ac_g da saqlangan qiymatlari va **z** chegaralari

. list ac_g z in 1/12

	ac_g	z
1.	.49425676	4.842708
2.	.4107073	4.024093
3.	.1544205	1.513006
4.	.20043788	1.963882
5.	.09038538	.8855922
6.	.02447111	.239767
7.	-.03008434	-.2947652
8.	-.08231978	-.8065658
9.	.04410661	.4321548
10.	-.02128483	-.2085479
11.	-.08683463	-.8508022
12.	-.20404326	-1.999207

Phillips egri chizig'i

Ikkinci misol Fillips egri chizig'iga asoslangan bo'lib, inflatsiya va ishsizlik orasidagi munosabatni ifodalaydi.

Inflyatsiya va ishsizlikning o'zgarishi bilan bog'liq oddiy regressiya:

$$INF_t = \beta_1 + \beta_2 DU_t + e_t$$

Model Avstraliyaning inflyatsiya darajasi va ishsizlik darajasini 1987yil 1-chorakni o'z ichiga olgan *phillips_aus.dta* ma'lumotlari yordamida baholanadi. Ma'lumotlarni yuklang, sanani satrga formatlang va ma'lumotlar to'plamini vaqtli qatorlar sifatida o'rnatting.

use phillips_aus, clear

generate date = tq(1987q1) + _n-1


format %tq date

tsset date

Birinchidan, inflyatsiya darajasi va ishsizlikning o'zgarishini grafiklarda ifodalang.

tsline inf

tsline D.u



Keyin eng kichik kvadratlar yordamida modelni baholang va qoldiqni toping.

reg inf D.u


predict ehat, res

reg inf D.u					
Source	SS	df	MS	Number of obs = 90	
Model	2.04834633	1	2.04834633	F(1, 88) = 5.29	
Residual	34.0445426	88	.386869802	Prob > F = 0.0238	
Total	36.0928889	89	.405538077	R-squared = 0.0568	
inf	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
D1.	-.5278638	.2294049	-2.30	0.024	-.9837578 - .0719699
_cons	.7776213	.0658249	11.81	0.000	.646808 .9084345

Qoldiqlar **residual correlogram** yordamida avtokorrelyatsiya uchun tekshiriladi. Qoldiq korrelogramma e_t va e_{t-1} o`rtasidagi avtokorrelatsiyalar qatorini kuzatishlar orasidagi vaqt oralig`iga nisbatan $j=1,2,\dots,m$ ga teng chiziqli grafik ifodalanadi. Namunadagi avtokorrelyatsiyalar rk deb nomlangan o`zgaruvchida saqlanadi, birinchi beshtasi chop etiladi va keyin ma`lumotlar to`plamidan o`chiriladi, chunki ular endi kerak emas.

ac ehat, lags(12) generate(rk)

list rk in 1/5



. list rk in 1/5

	rk
1.	.54865864
2.	.45573248
3.	.43321579
4.	.42049358
5.	.33903419

Ko'rinib turibdiki, bir qator muhim avtokorrelyatsiyalar mavjud va ular nisbatan katta. Stata **corrgram** funktsiyasini o'z ichiga oladi, uning kichik to'plami **ac** buyrug'i hisoblanadi. **Corrgram** avtokorrelyatsiyalar jadvalini ishlab chiqaradi (shuningdek, qisman avtokorrelyatsiyalar va Portmanto (Q) statistikasi). Shuningdek, u avtokorrelyatsiyalarning xarakterga asoslangan maydonini ko'rsatadi. **Corrgram** ning yana bir xususiyati bu statistikaning har biri **r()** sifatida saqlanadi. Birinchi beshlik avtokorrelyatsiyani saqlash va natijani olish uchun **corrgram** dan foydalilaniladi.

corrgram ehat, lags(5)

. corrgram ehat, lags(5)

LAG	AC	PAC	Q	Prob>Q	-1 [Autocorrelation]	0 [Autocorrelation]	1 [Partial Autocor]	0 [Partial Autocor]	1 [Partial Autocor]
1	0.5487	0.5498	28.006	0.0000					
2	0.4557	0.2297	47.548	0.0000					
3	0.4332	0.1926	65.409	0.0000					
4	0.4205	0.1637	82.433	0.0000					
5	0.3390	0.0234	93.63	0.0000					

Birinchi uchta avtokorrelyatsiya namunasini natijani olish:

. di "rho1 = " r(ac1) " rho2 = " r(ac2) " rho3 = " r(ac3)
rho1 = .54865864 rho2 = .45573248 rho3 = .43321579

corrgram bajargandan so'ng saqlanadigan boshqa statistik ma'lumotlarni ko'rish uchun **return list** kiriting.

9.4. Ketma-ket korelyatsiya uchun boshqa testlar

Ketma - ket korrelyatsiya uchun ikkinchi test 8-bobda heteroskedastiklik kontekstida muhokama qilingan *Langrage multiplicator testi* printsipiga asoslanadi. Sinov statistikasi TR^2 ga asoslangan yordamchi regressiya. Avtokorrelyatsiya uchun bu test yordamchi regressiyaga asoslanadi, bunda siz eng kichik kvadratlar qoldiqlari va asl regressorlar bo'yicha eng kichik qoldiqlarni regressiya qilasiz. Agar \hat{e}_{t-1} ni o'z ichiga olgan regressorlar

\hat{e}_t dagi yetarli o`zgarishlarni tushintirsa, u holda \hat{e}_{t-1} tufayli avtokorrelyastsiya bo`lishi kerak. Regressiya modeli uchun

$$y_t = \beta_1 + \beta_2 x_t + e_t$$

eng kichik kvadratlar yordamida parametrlarni baholang va qoldiqlarni saqlang \hat{e}_t . \hat{e}_{t-1} olish uchun Lag qoldiqlaridan foydalanamiz. Keyin, ikkinchi “yordamchi” regressiyani \hat{e}_t bog`liq o`zgaruvchi sifatida va kechikish qiymatini \hat{e}_{t-1} mustaqil o`zgaruvchi sifatida baholang. Asl regressiyadan boshqa barcha mustaqil o`zgaruvchilarni ham o`z ichiga oladi. Oddiy chiziqli regressiya uchun yordamchi regressiya hisoblanadi.

$$\hat{e}_t = \gamma_1 + \gamma_2 x_t + \rho \hat{e}_{t-1} + residual$$

Agar avtokorrelyatsiyaning yo`qligi haqidagi nol gipoteza to`g`ri bo`lsa, regressiyadan TR^2 , $x^2(1)$ taqsimatga ega bo`ladi, bu yerda T yordamchi regressiyadagi kuzatishlar soni. Muhim avtokorrelyatsiya mavjudligi haqidagi xulosa rad etishga olib keladi. Fillips egri chizig'i misoli uchun, *ehat* asl regressiyadan saqlanib qolgan deb faraz qilsak

quietly regress ehat D.u L.ehat

*di "Observations = " e(N) " and TR2 = " e(N)*e(r2)*

natija:

```
. di "Observations = " e(N) " and TR2 = " e(N)*e(r2)
observations = 89 and TR2 = 27.608808
```

Principles of Econometrics, 4-nashr adabiyotida bu statistika barcha mavjud kuzatishlar yordamida hisoblab chiqilgan. Modelda kechikkan(lagli) qiymat e_{t-1} paydo bo`lganligi sababli, yordamchi regressiyani baholashda odatda kuzatuv kamayadi. Bu holda, birinchi uchun yetishmayotgan qiymat qoldiq nolga almashtirilishi mumkin. Bu joriy kontekstda ruxsat etiladi, chunki uning kutilgan qiymati shunday (ya`ni, $E(e_1)=0$). To`gri test statistikasini olish uchun texnik jihatdan $e_1=0$ o`rnatish kerak emas; ammo, natijani matnda takrorlash uchun bu yerda amalga oshiriladi. Keyin,

replace ehat = 0 in I

quietly regress inf D.u L.ehat

*di "Observations = " e(N) " and TR2 = " e(N)*e(r2)*

natijada esa:

```
. di "Observations = " e(N) " and TR2 = " e(N)*e(r2)
observations = 90 and TR2 = 27.592347
```

Bu matndagi natijaga mos keladi. Yuqori darajadagi avtokorrelyatsiyani tekshirish oddiy. AR(4) ni hisoblash uchun eng kichik kvadratlar qoldiqlarini regressorlar sifatida kiriting va TR^2 ni hisoblang. Chi-kvadrat uchun erkinlik darajalari muqobil ostidagi avtokorrelyatsiya tartibiga teng (bu holda, 4).

Buni amalga oshirish uchun buyruq:

quietly regress ehat D.u L(1/4).ehat

natijada:

```
. di "Observations = " e(N) " and TR2 = " e(N)*e(r2)
observations = 86 and TR2 = 33.385269
```

POE4 adabiyotida natijalarni qo'lda takrorlash biroz ishni talab qiladi. Kechikishlarni qabul qilishda yuzaga keladigan *ehat* yetishmayotgan qiymatlari nolga o`rnataladi (ya`ni, $e_3=0$, $e_2=0$, $e_1=0$, $e_0=0$). Bu butun namunadan foydalanish imkonini beradi. Ma'lum bo'lishicha, bu Stata-da dasturlash juda oson emas, shuning uchun biz bu yerda muhokamani o'tkazib yuboramiz biroq, buni amalga oshirish uchun buyruqni ushbu bobning oxiridagi do-faylda topish mumkin.

Natijalarni *estat bgodfrey* o`rnatalgan post- baholash buyrug`i yordamida osongina takrorlash mumkin.

regress inf D.u

estat bgodfrey, lags(1)

estat bgodfrey, lags(4)

Buyruq modeldagи regressorlar sifatida qancha kechikkan qoldiqlarni kiritish kerakligini ko`rsatadigan variantdan foydalanadi. AR(1) misolida, muqobil gipoteza, modelning xatolari birinchi tartibli avtokorrelyatsiyaga ega; kechikishlar(1) ishlataladi. Natijada:

```
. estat bgodfrey, lags(1)
Breusch-Godfrey LM test for autocorrelation
-----
```

lags(ρ)	chi2	df	Prob > chi2
1	27.592	1	0.0000

H0: no serial correlation

AR(4) muqobil uchun

```
. estat bgodfrey, lags(4)
Breusch-Godfrey LM test for autocorrelation
-----
```

lags(ρ)	chi2	df	Prob > chi2
4	36.672	4	0.0000

H0: no serial correlation

Har bir holatda oddiy regressiya qoldiqlarida avtokorrelyatsiyaning aniq natijalari mavjud.

9.5. Ketma-ket bo`g`liq xatolar bilan baholash

Regressiya modeli regressorlar sifatida qaram o`zgaruvchining kechikishlarini o`z ichiga olmasa, eng kichik kvadratlar xatolar AR(q) modeliga amal qilgan taqdirda ham izchil bo`ladi. $t \neq s$ uchun MR4, $cove(e_t, e_s) = 0$ eng kichik kvadratlar farazi buzilganda u endi samarali emas (asimptotik). Afsuski, odatiy standart xatolar endi to`g`ri emas, bu statistik jihatdan noto`g`ri gipoteza testlari va ishonch oralig`iga olib keladi.

9.5.1. Eng kichik kvadratlar va HAC standart xatolar

Odatdagidek eng kichik kvadratlar standart xatolari, xuddi biz heteroskedastik modellarda bo`lgani kabi, Newey va West tomonidan baholovchi yordamida taklif qilingan.

Newey-West standart xatolar (HAC deb ham ataladi – geteroskdastiklik va avtokorrelyatsiya izchil standart xatolar) 8-bobda kiritilgan geteroskdastik izchil standart xatolarga o`xshaydi. Ular avtokorrelyatsiya qilingan xatolar uchun AR(1) bo`lishi shart bo`lman va pastroq dispersiyaga ega bo`lgan baholovchini olish uchun zarur bo`lgan dinamik xato modelining o`ziga xosligini talab qilmaydigan izchil bo`lish afzalligiga ega. HAC 8-bobdagagi heteroskedastika mustahkam standart xato hisoblagichi kabi avtomatik ishlatilmaydi. Avtokorrelyatsiyaga nisbatan ishonchli bo`lishi uchun qoldiq avtokorrelyatsiya vaqt o`tishi bilan qanchalik muhim bo`lishi mumkinligini aniqlash kerak. Asosan, tanlangan vaqt oynasidagi avtokorrelyatsiya qilingan xatolar HAC hisoblashda o`rtacha hisoblanadi; o`rtacha hisoblab chiqiladigan davrlar soni va har bir qoldiq uchun qancha og`irlilik berilishi foydalanuvchi tomonidan belgilanishi kerak.

O`rtacha og`irlilik **kernel** deb ataladigan narsa yordamida amalga oshiriladi va tortish sxemasi (yadrosoi) yordamida o`rtacha xatoliklar soni **bandwidth** deb ataladi. Rostini aytasam, bu atamalar o`rtacha foydalanuvchi haqida ozgina ma`lumot beradi. Yadroni o`rtacha og`irlilikdagi boshqa nom va tarmoqli kengligini o`rtacha atamalar soni uchun atama sifatida o`ylab ko`ring. Stata kernel tanlashning hech qanday usulini taklif qilmaydi; yagona **Bartlett** mavjud. Biroq, **bandwidth** kengligi tanlanishi kerak.

Mos **bandwidth** kengligi tanlashga yordam beradigan bir necha usullar mavjud va ikkitasi bu yerda berilgan. T, ikkala holatda ham **bandwidth** kengligi namuna hajmiga bog`liq. Birinchisi $B=0,75T^{1/3}$ dan foydalanadi. Boshqa mashhur tanlov- $B=4(T/100)^{2/9}$. Bu EViews kabi boshqa dasturlarda standart bo`lib ko`rinadi va bu yerda matndagi natijalarni olish uchun foydalaniladi.

Bilvosita ko`rib chiqilishi kerak bo`lgan kelishuv mavjud. Kattaroq **bandwidth** kengligi notog`rilikni (yaxshi) va aniqlikni (yomon) kamaytiradi. Kichikroq **bandwidth** kengligi ko`proq mos keladigan avtpkorrelyatsiyalarni istisno qiladi (shuning uchun koproq tarafkashlik qiladi), lekin kichikroq farqqa ega. Umumiylashtirish eng katta avtokorrelyatsiyalarni o`z ichiga oladigan darajada katta bo`lgan tarmoqli (**bandwidth**) kengligini tanlashdir.

O`tkazish bandwidth ni hisoblash uchun quyidagi buyruqdan foydalaning:

$$\text{scalar } B = \text{round}(4 * (e(N)/100)^{(2/9)})$$

scalar list B

Bu *philips _ aus.dta* ma`lumotlar to`plamidagi 4 qiymatini qaytaradi. Natija yaxlitlanadi, chunki Stata HAC ning hisoblashdagi keshikishlar sonini belgilash uchun butun sondan foydalanishni talab qiladi.

Stata-da mavjud bo`lgan yagona kernel bu Bartlett dir. Bu Newey va West tomonidan ushbu masala bo`yicha tadqiqotlarida foydalanilgan. Shuning uchun Stata HAC ni newey sifatida hisoblaydigan jarayonga ishora qiladi. U asosan regress o`rnini bosadi va u tarmoqli kengligi spetsifikatsiyasini talab qiladi. Keyin modelni Newey-West standart xatoliklari va

tarmoqli kengligi 4 bo`lgan eng kichik kvadratlar bo`yicha baqholash uchun quyidagi buyruqdan foydalaning.

newey inf D.u, lag(4)

Misolda model odatdagи eng kichik kvadratlar standart xatolari va HAC standart xatolari bilan eng kichik kvadratlar yordamida baholanadi. Natijalar quyida ko`rinadi, HAC standart xatolar o`ng tomondagi ustindagi taxminlar ostida ko`rinadi.

```
esttab Wrong_SE HAC_4, compress se(%12.3f) b(%12.5f) gaps ///
scalars(r2_a rss aic) title("Dependent Variable: inf") ///
mtitles("LS" "HAC(4)")
```

Dependent variable: inf		
	(1) LS	(2) HAC(4)
D.u	-0.52786* (0.229)	-0.52786 (0.318)
_cons	0.77762*** (0.066)	0.77762*** (0.112)
N	90	90
r2_a	0.04603	
rss	34.04454	
aic	171.91634	.

Standard errors in parentheses
* p<0.05, ** p<0.01, *** p<0.001

Compress opsiysi qatorlar orasidagi vertical bo`shliqni qisqartirish uchun ishlatiladi, gaps opsiysi koeffitsiyentlar orasiga bo`sh qatorlarni (yoki umuman olganda, qo`shimcha vertical bo`shliq) qo`sadi va scalars opsiysi turli statistik ma`lumotlarni ifodalash imkonini beradi, regressiya natijalari bilan birga saqlanadi. Bu holda, sozlangan R^2 , kvadratlar xatosining regressiya yig`indisi va Stata tomonidan AIC mezonini hisoblash. Bundan tashqari, *mttitle* variantlari har bir ustinga mazmunli nom berish uchun ishlatiladi; bu parameter qo`llanilganda, qaram o`zgaruvchining nomi bo`lgan standart ustun nomi har bir qo`shtirnoq to`plamiga qo`ygan narsangiz bilan almashtiriladi. Title variant o`quvchilarga har bir holatda ishlatiladigan qaram o`zgaruvchi *inf* ekanligini bilish uchun ishlatiladi. Bu misolda, HAC standart xatolar odatdagidan (mos kelmaydigan) sezilarli darajada kattaroqdir.

9.5.2. Nochiziqli eng kichik kvadratlar

Ko`rib turganingizdek, HAC standart xatolar kamida ikkita kamchilikka ega: 1) ular avtomatik emas, chunki ular tarmoqli kengligi spetsifikatsiyasini talab qiladi; 2) ular oddiy chiziqli regressiyaga qaraganda samaraliroq bo`lgan ko`proq baholovchilarning standart xatolaridan kattaroqdir. Ushbu bo`limda AR(1) modelining parametrlarini samarali baholash uchun chiziqli bo`lmagan eng kichik kvadratlardan foydalilanildi.

Sizning darsligingizda mualiflar AR(1) regressiya modelidan boshlaydilar va bir oz algebradan foydalaniib, bu natjalarga erishiladi.

$$y_t = \beta_1(1 - \rho) + \beta_2 x_t + \rho y_{t-1} - \rho \beta_2 x_{t-1} + v_t$$

Bu model parametrlarda chiziqli emas, lekin qo'shimcha oq shovqin xatosiga ega. Ushbu xususiyatlar modelni chiziqli bo'lмаган eng kichik kvadratlarni baholash uchun mos qiladi. Chiziqli bo'lмаган eng kichik kvadratlar kvadrat xatolar yig'indisini minimallashtiradigan parametrlearning qiymatlarini topish uchun raqamli usullardan foydalanadi. Modelni baholash uchun Stataning umumiyligi chiziqli bo'lмаган eng kichik kvadratlar buyrug'idan foydalaning, nl:

***nl (inf = {b1}*(1-{rho}) + {b2}*D.u + {rho}*L.inf - {rho}*{b2}*(L.D.u)), ///
variables(inf D.u L.inf L.D.u)***

Sintaksis juda oddiy, ammo ba'zi tushuntirishlarni talab qiladi. Asosiy sintaksis quyidagicha:

nl (depvar=) [if] [in] [weight] [, options]

Modelning tizimli qismi birinchi qavslar to'plamiga kiritilgan. Parametrlar qavslar {} ichiga olinishi kerak. If, in and weight iboralari chiziqli regressiyadagi kabi ishlataladi. Biroq, kechikkan o'zgaruvchilar tufayli, ma'lumotlar to'plamidagi kechikkan o'zgaruvchilar bo'yicha birinchi kuzatish uchun yetishmayotgan qiymatlar yaratiladi. Buning ishlashi uchun namuna faqat to'liq kuzatuvlar bilan cheklanishi kerak. Buning ikkita usuli bor bu. Birinchidan, siz (depvar=<sexp>) 2/34 da foydalanishingiz mumkin. Yoki o'zgaruvchilarni bajarilganidek ro'yxatlashingiz mumkin bu yerda variant o'zgaruvchilari (inf D.u L.inf L.D.u) yordamida topiladi.

Baholash natijalari:

Source	SS	df	MS	Number of obs	=	89
Model	12.3860433	2	6.19302165	R-squared	=	0.3481
Residual	23.1986758	86	.269752044	Adj R-squared	=	0.3329
Total	35.5847191	88	.404371808	Root MSE	=	.5193766
				Res. dev.	=	132.9069

inf	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
/b1	.7608716	.1245311	6.11	0.000	.513312 1.008431
/rho	.5573922	.0901546	6.18	0.000	.3781709 .7366136
/b2	-.694388	.247894	-2.80	0.006	-1.187185 -.201591

Parameter b1 taken as constant term in model & ANOVA table

Koeffitsient baholari matndagilarga mos keladi. Xuddi shu parameter baholarida funktsiya kvadratlari yig'indisining minimaliga erishiladi. Hisoblangan standart xatolarda ba'zi kichik farqlar mavjud. Bu chiziqsiz modellarda ularni izchil ravishda baholashning turli usullari mavjudligi sababli sodir bo`ladi; ushbu misoldagi kabi kichik namunalarda bu farqlar bo`rtirilgan bo`lishi mumkin. Kattaroq namunalarda farqlar odatda kichik bo`ladi va namuna hajmi o'sishi bilan nazariyaga ko`ra yo`qoladi. Parametr ρ bo'yicha t-nisbati 6.18 ga teng, bu p-qimati 0,001 dan kam. Bu shuni anglatadiki, har qanday o'rtacha darajadagi ahamiyatga ega (masalan 5%) qoldiqlar o'rtasida birinchi tartibli avtokorrelyatsiya dalillari mavjud. Modelni baholagandan so'ng, keyingi bo`limda foydalanish uchun bir nechta skalerlar hisoblab chiqiladi.

```

scalar delta = _b[b1:_cons]*(1-_b[rho:_cons])
scalar delta1 = -_b[rho:_cons]*_b[b2:_cons]

```

delta deb nomlangan skalyar $\hat{\beta}_1(1 - \hat{\rho})$ va **delta1** $-\hat{\rho}\beta_2$. Bularning sabablari keyingi bobda muhokama qilinadi. Biroq e`tibor bering, taxminlar chiziqli regressiyaga qaraganda biroz boshqacha ko`rsatilgan. Chiziqli modellarda ishlataladigan **_b[varname]** konvensiyasi **_b[paramname:_cons]** bilan almashtiriladi. Parametrler uchun tegishli nomlarni toppish uchun **nl** buyrug`idan keyin **coflegend** variantidan foydalanish mumkin. Parametrлари то`г`ри aniqlaganligingizni tekshirib ko`ring, **coflegend** buyrug`и yordamida chiziqli bo`lmagan eng kichik kvadratlar regressiyasini qayta ishga tushiring. Bu odatda kerakli natijaning ko`p qismini bostiradi, lekin u har bir parameter uchun Stata nomlarini aniqlaydigan afsonani yaratadi. Bobning oxiridagi do-faylda buning misoli mavjud.

9.5.3. Odatda ko`proq ishlataladigan model

Modelning umumiyroq shakli ko`rib chiqamiz.

$$y_t = \delta + \delta_0 x_t + \delta_1 x_{t-1} + \theta_1 y_{t-1} + v_t$$

Parametrлarda chiziqli va chiziqli regressiya bilan baholanishi mumkin. Bu model oldingi modelga munosabatlar orqali bog`langan.

$$\delta = \beta_1(1 - \rho) \quad \delta_0 = \beta_2 \quad \delta_1 = -\rho\beta_2 \quad \theta_1 = \rho$$

Chiziqli modelni (chiziqli) eng kichik kvadratlar bilan baholash mumkin va nazarda tutilgan chekloving gipoteza testini o`tkazish mumkin. Cheklovlar nazarda tutuvchi nol gipoteza $H_0: \delta_1 = \theta_1 \delta_0$ teng emas degan muqobilga qarshi gipotezadir. Birinchi qadam eng kichik kvadratlar yordamida modelni baholashdir.

regress inf L.inf D.u L.D.u

. regress inf L.inf D.u L.D.u						
Source	SS	df	MS	Number of obs = 89 F(3, 85) = 15.18 Prob > F = 0.0000 R-squared = 0.3489 Adj R-squared = 0.3260 Root MSE = .52208		
Model	12.4166337	3	4.13887791			
Residual	23.1680854	85	.27256571			
Total	35.5847191	88	.404371808			
inf	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
inf L1.	.5592676	.0907962	6.16	0.000	.3787403	.7397948
D1. LD.	-.6881852 .3199526	.2498704 .257504	-2.75 1.24	0.007 0.217	-1.184994 -.1920343	-.191376 .8319396
_cons	.3336325	.0899028	3.71	0.000	.1548817	.5123834

Oldingi bo`lim oxirida hisoblangan skalyar δ va δ_1 ga to`g`ri keladi. Hisoblangan qiymatlar edi

```

. scalar list delta delta1
    delta = .33676767
    delta1 = .38704645

```

Chiziqli regressidan olingen umumiylar $\delta=0,334$ va $\delta_1=0,320$. Ikkala qiymat ham ***nl*** tomonidan baholangan cheklovchi modelda nazarda tutilgan qiymatlarga juda yaqin.

Chiziqsiz gipotezani haqiqatda sinab ko`rish uchun $H_0: \delta_1 = -\theta \delta_0$ parametrarning chiziqli bo`lmagan funktsiyasini sinab ko`rish uchun Stataning o`rnatilgan funktsiyasidan foydalaning, ***testnl***.

testnl _b[L.D.u]=-_b[L.inf]*_b[D.u]

```
. testnl _b[L.D.u]=-_b[L.inf]*_b[D.u]
(1) _b[L.D.u] = -_b[L.inf]*_b[D.u]
F(1, 85) =      0.11
Prob > F =      0.7384
```

Katta p-qiyomi 0,74 chiziqli bo`lmagan eng kichik kvadratlar bilan baholangan AR(1) modeli haddan tashqari cheklovchi emasligini ko`rsatadi.

esttab buyrug`i yordamida ko`rib chiqilayotgan modellarning turli chiziqli xususiyatlari taqqoslanadi.

```
> esttab General No_LDu Original, compress se(%12.3f) b(%12.5f) ///
>           gaps scalars(r2_a rss aic)
```

	(1) inf	(2) inf	(3) inf
L.inf	0.55927*** (0.091)	0.52825*** (0.085)	
D.u	-0.68819** (0.250)	-0.49086* (0.192)	-0.52786* (0.229)
LD.u	0.31995 (0.258)		
_cons	0.33363*** (0.090)	0.35480*** (0.088)	0.77762*** (0.066)
N	89	90	90
r2_a	0.32595	0.33137	0.04603
rss	23.16809	23.59054	34.04454
aic	140.78946	140.90217	171.91634

Standard errors in parentheses
* p<0.05, ** p<0.01, *** p<0.001

compress buyrug`i qatorlar orasidagi vertical bo`shliqni qisqartirish uchun ishlataladi, gaps opsiyasi koeffitsientlar orasiga bo`sh qatorlarni (yoki umuman olganda, qo`shimcha vertikal bo`shliqni) qo`shadi va scalars opsiyasi saqlanadigan turli statistik ma`lumotlarni chop etish imkonini regressiya natijalari bilan birga beradi. Bu holda, sozlangan R^2 kvadratlar xatosining regressiya yig`indisi va Stata tomonidan AIC mezonini hisoblash kerak.

9.6. Avtoregressive taqsimlangan kechikish(lagli) modellari

Cheklangan taqsimlangan kechikishlarni birlashtirgan va avtoregressiv model ko`rib chiqiladi. Bu avtoregressiv taqsimlangan kechikish modeli (ARDL) deb ataladi. ARDL (p,q) modeli umumiylar shaklga ega:

$$y_t = \delta + \theta_1 y_{t-1} + \dots + \theta_p y_{t-p} + \delta_0 x_t + \delta_1 x_{t-1} + \dots + \delta_q x_{t-q} + v_t$$

Regressor sifatida u p bog`liq o`zgaruvchining y_t va q mustaqil o`zgaruvchining x_t kechikishiga ega. Inflyatsianing ARLD (1,1) va ARLD (1,0) modellari eng kichik kvadratlar yordamida baholaymiz. Taxminlar saqlanadi va quyidagi jadvalda natijalar ifodalaniladi.

regress inf L.inf L(0/1).D.u

estimates store AR1_DL1

regress inf L.inf D.u

estimates store AR1_DL0

```
. esttab AR1_DL1 AR1_DL0, compress se(%12.3f) b(%12.5f) ///
> gaps scalars(r2_a rss aic)
```

	(1)	(2)
	inf	inf
L.inf	0.55927*** (0.091)	0.52825*** (0.085)
D.u	-0.68819** (0.250)	-0.49086* (0.192)
LD.u	0.31995 (0.258)	
_cons	0.33363*** (0.090)	0.35480*** (0.088)
N	89	90
r2_a	0.32595	0.33137
rss	23.16809	23.59054
aic	140.78946	140.90217

Standard errors in parentheses

* p<0.05, ** p<0.01, *** p<0.001

Ushbu modellar orasidan tanlash bir necha usul bilan amalgalash mumkin. Birinchidan, agar DU_{t-1} dagi t-ratio ahamiyatsiz bo`lsa unda dalillar shuni ko`rsatadiki, uni o`tkazib yuborish cheklangan modelning eng kam chayqaladigan baholovchisi xususiyatlarga salbiy ta`sir ko`rsatmasligi mumkin.

Yana bir imkoniyat - 6-bobda muhokama qilingan model tanlash qoidalariidan birini qo'llash mumkin. Eslaymizki, biz AIC va SC modelini tanlash qoidalari hisoblaydigan ***modelsel*** deb nomlangan dasturni yozgan edik. Bu yerda dastur hisoblangan R^2 displayini qoldirib, modeldagи kuzatuvlar sonini chop etish orqali biroz o`zgartiriladi. Stata-da dastur tuzilishi haqida bat afsil ma'lumot olish uchun 6-bobga qarang. AIC yoki SC yordamida ARDL (1,1) va ARDL (1,0) o`rtasida tanlash uchun ***modelsel*** deb nomlangan quyidagi dasturni yaratishing va ishga tushiring.

program modelsel

scalar aic = ln(e(rss)/e(N))+2*e(rank)/e(N)

scalar sc = ln(e(rss)/e(N))+e(rank)*ln(e(N))/e(N)

scalar obs = e(N)

scalar list aic sc obs

end

Endi quyida ko'rsatilgan tanlov mezonlarini tekshirib, har bir modelni baholang. Bu quyidagi natijalarni chiqaradi:

```

. quietly regress inf L.inf L(0/1).D.u
. modelsel
  aic = -1.255973
  sc = -1.1441242
  obs = 89

. quietly regress inf L.inf L.D.u
. modelsel
  aic = -1.1929642
  sc = -1.1090776
  obs = 89

```

ARDL (1,0) ham AIC, ham SC ni kamaytiradi va afzal qilingan modeldir. Ushbu tahlil bilan bog`liq muammolardan biri shundaki, qoldiqlar hali ham avtokorrelyatsiyalangan bo`lishi mumkin yoki bu yerda ko`rib chiqilganlardan uzoqroq kechikishlar o`tkazib yuborilgan. Keyingi bo`limda bu batafsilroq ko`rib chiqiladi.


9.6.1. Fillips egri chizig'i

Birinchidan, ARDL (1,0) xatolari avtokorrelyatsiya uchun tekshirilishi kerak. Buni korrelogrammaga qarash orqali amalga oshirish mumkin.

quietly regress inf L.inf D.u

predict ehat, res

ac ehat, lags(12)



yoki LM (Breusch-Godfrey) testi bo`yicha, bu holda 1-5 kechikishlar uchun statistik ma`lumotlarni o`z ichiga oladi.

. estat bgodfrey, lags(1 2 3 4 5)

Breusch-Godfrey LM test for autocorrelation

lags(<i>p</i>)	chi2	df	Prob > chi2
1	4.130	1	0.0421
2	5.123	2	0.0772
3	5.221	3	0.1563
4	9.554	4	0.0486
5	12.485	5	0.0287

H0: no serial correlation

Ikki jarayon natijalari ARDL (1,0) qoldiqlarida avtokorrelyatsiya mavjudligi haqida turli xil taassurotlarni beradi. Korrelogrammdagi avtokorrelyatsiyalarning hech biri 95% ishonch oralig`idan tashqarida emas, lekin 1,4 yoki 5 kechikishlarga ega modellar uchun LM statistikasi 5% darajasida statistik ahamiyatga ega. Ikkinchidan model tanlash qoidalarini ko`proq avtokorrelyatsiya shartlarini o`z ichiga olgan kengroq modellarga qo'llash kerakligini taklif qiladi. Bu o'n ikkita modelni baholashni anglatadi; AR shartlari 1 dan 6 gacha, DL esa 0 dan 1 gacha, har bir taxminiy kombinatsiya bilan farqlanadi. AIC va SC statistikasi har bir model uchun hisoblab chiqiladi va umumiy minimallar qidiriladi. 1988q3 dan keyingi davrlar uchun ushbu modellarning har birini baholash uchun Stata buyrug`i ushbu bobning oxirida do-faylda keltirilgan. To`liq buyruq POE4 adabiyotning 9.4-jadvalida keltirilgan natijani takrorlash uchun ishlatalishi mumkin. Quyida buyruq parchasi berilgan va uning sintaksisi tushintirilgan.

Quyidagi buyruq 1983 yilning uchinchi choragida boshlangan ma`lumotlar uchun ARDL (1,1) modelini baholaydi. Regressiya haqiqiy regressiya natijalarini bostirish uchun Stata quyida qisqartirilgan *quietly* buyrug`i yordamida baholanadi; bizning qiziqishimiz faqat shu nuqtada model tanlash qoidalarining qadriyatlariga bog`liq. Namunani ma`lum sanalar bilan cheklash uchun psevdofunktsiya *tq(1988q3)* ishlataladi. Eslatib o`tamiz, ushbu bobning boshida ushbu psevdofunktsiya 1988q3 sanasini Stata tushinadigan raqamga aylantirilgandi. Choraklik ma`lumotlar uchun bu to`g`ri sintaksisdir.

```
qui reg L(0/1).inf L(0/1).D.u if date>= tq(1988q3)
di "p=1 q=1"
modelsel
```

Kechikish operatorlari bitta bayonotda **L(0/1).inf**. qaram o`zgaruvchi va avtoregressiv mustaqil o`zgaruvchilarni ko`rsatish orqali o`zlarining to`liq afzalliklaridan foydalanilmoqda. Ushbu bayonotda birinchi o`zgaruvchi inflyatsiyaning nol kechikishi **L(0).inf** bo`lib, u shunchaki *inf*. Demak, **L(0/1).inf** iborasi *inf L.inf* ga ekvivalentdir, chunki *inf* regress dan keyin birinchi bo`lib paydo bo`ladi, Stata uni bog`liq o`zgaruvchi sifatida tan oladi.

Bu buyruq qismidan olingan natijalar

```
. di "p=1 q=0"
p=1 q=0

. modelsel
    aic = -1.2466292
    sc = -1.160418
    obs = 85
```

Halqalardan foydalanish model tanlashni ancha osonlashtirishi mimkin. Misol sifatida, keling, $p=1,2,3,4,5$ va 6 $q=0$ va 1 uchun barcha mumkin bo`lgan modellarni qidiramiz. Ketma-ket qiymatlar ustida aylanadigan **forvalues** buyrug`i yordamida ichki halqa hosil qilish mumkin.

Asosiy tuzilma quyidagicha bo`ladi:

```

forvalues q=0/1 {
    forvalues p=1/6 {
        [statements to compute and print]
    }
}

```

q va p ning hisoblangan qiymatlari berilgan diapazonda bo`lsa, tsikillar bajariladi (masalan, q uchun 0 va 1 p uchun 1dan 6 gacha). Bu shaklda p va q qiymatlari 1 bosqichga ortadi. Qavslar *forvalues* bilan belgilanishi kerak va

1. ochiq qavs *forvalues* bilan bir qatorda paydo bo`lishi kerak;
2. ochiq qavsga, albatta, izohlardan tashqari hech narsa amal qilolmaydi; bajariladigan birinchi buyruq yangi satrda paydo bo`lishi kerak;
3. yaqin qavs o`z-o`zidan chiziqda paydo bo`lishi kerak.

Okun qonunining ARDL (p,q) uchun buyrug`i quyidagicha ko`rinadi.

```

forvalues q=0/1 {
    forvalues p=1/6 {
        quietly regress L(0/^p').inf L(0/^q').D.u if date >= tq(1988q3)
        display "p=`p' q=`q'"
        modelsel
    }
}

```

Halqalar ichida hisoblanayotgan bayonotlar haqida bir nechta narsaga e`tibor bering. Birinchidan, p va q endi makro nomlari bilan ataladi. Bu shuni anglatadiki, ular haqida gap ketganda, ular bitta qo`shtirnoq ichiga olinishi kerak (yuqorida qilganimizdek, chap va o`ng). Ikinchidan, bog`liq o`zgaruvchi va avtoregressiv mustaqil o`zgaruvchi yana bitta *L(0/^p').inf* bayonotiga kiritilgan. p soni 1dan 6 gacha oshib, kechikishlar qo`shiladi va p va q ning ekranga chop etilgandan so`ng *modelsel* dastur bajariladi. Shunday qilib, bir nechta qisqa bayonotlarda ko`plab modellarni ko`rib chiqish mumkin va taqsimlangan kechikishlar tartibini osongina o`zgartirish mumkin. Tsikillar shu tarzda o`rnatilganda, q tsikili noldan boshlanadi, keyin p tsikili 1 dan 6 gacha takrorlanadi. Birida p tsikili tugaydi, q tsikili 1 ga ortadi va p tsikili qaytadan boshlanadi. Agar hoxlasangiz, ularning tartibini o`zgartirishingiz mumkin.

9.6.2. Okun qonuni

Okun qonuni vaqtli qatorlar modelining adekvat spetsifikatsiyasini izlash uchun yana bir imkon beradi. *okun.dta* ma`lumotlarini yuklang, 1985q2 dan boshlangan sanalarni yarating, ularni satrlar sifatida chop etish uchun formatlang va ma`lumotlarni vaqtli qatorlar deb e`lon qiling.

```

use okun, clear
generate date = tq(1985q2) + _n-1
format %tq date

```

tsset date

9.2-bo`limda taxmin qilingan model ARDL (0,2) bo`lib, ishsizlik darajasining o`zgarishini YAIM o`sishi bilan bog`laydi. Model ostida eng kichik kvadratlar bilan baholanadi, korrelogramm olinadi va 5 tagacha avtokorrelyatsiya qilingan qoldiqlarni o`z ichiga olgan modellar uchun LM statistikasi ishlab chiqariladi.


reg D.u g L(1/2).g L.D.u

predict ehat, res

ac ehat, lags(12)

drop ehat

estat bgodfrey, lags(1 2 3 4 5)



. estat bgodfrey, lags(1 2 3 4 5)

Breusch-Godfrey LM test for autocorrelation

lags(ρ)	chi2	df	Prob > chi2
1	12.364	1	0.0004
2	12.894	2	0.0016
3	13.754	3	0.0033
4	15.228	4	0.0043
5	19.648	5	0.0015

H0: no serial correlation

Korrelogramm bitta muhim avtokorrelyatsiyaga ega va LM statistikasining har biri 5% darajasida ahamiyatlidir. Bu, ARDL (0,2) noto`g`ri ko`rsatilganligini ko`rsatadi. Model oxiridagi do-faylda Okun ma`lumotlar to`plamidan foydalangan holda bir qator modellarni baholash uchun buyruq berilgan. Namuna oldingi misoldagidek cheklangan, bu safar 1986 yilning birinchi choragida boshlangan kuzatishlar uchun.

Avtokorrelyatsiyalar 0 dan 2 gacha va taqsimlangan kechikishlar 1 dan 3 gacha o`zgaradi; bu taxmin qilish uchun 9 ta modelga olib keladi.

forvalues q=1/3 {

forvalues p=0/2 {

quietly regress L(0/^p').D.u L(0/^q').g if date >= tq(1986q1)

display "p=`p' q=`q'"

modelsel

```
}
```

Bu quyidagini hosil qiladi:

```
p=0  q=1
    aic = -3.4362364
    sc = -3.3555876
    obs =      95
p=1  q=1
    aic = -3.5879866
    sc = -3.480455
    obs =      95
p=2  q=1
    aic = -3.5693074
    sc = -3.4348928
    obs =      95
p=0  q=2
    aic = -3.4633827
    sc = -3.355851
    obs =      95
p=1  q=2
    aic = -3.5675498
    sc = -3.4331352
    obs =      95
p=2  q=2
    aic = -3.5483196
    sc = -3.3870221
    obs =      95
p=0  q=3
    aic = -3.4424223
    sc = -3.3080077
    obs =      95
p=1  q=3
    aic = -3.5611594
    sc = -3.3998619
    obs =      95
p=2  q=3
    aic = -3.5490965
    sc = -3.3609161
    obs =      95
```

AIC va SC ni minimallashtiradigan model ARDL (1,1). Ushbu model butun namunadan foydalangan holda baholanadi va xatolar LM statistikasi yordamida qolgan avtokorrelyatsiya uchun tekshiriladi.

Quyidagi buyruqlar yordamida amalga oshiriladi:

**reg D.u L.D.u L(0/1).g
estat bgodfrey**

Natija:

Source	SS	df	MS	Number of obs = 96 F(3, 92) = 69.58 Prob > F = 0.0000 R-squared = 0.6941 Adj R-squared = 0.6841 Root MSE = .16228		
Model	5.49727601	3	1.83242534			
Residual	2.42272399	92	.026333956			
Total	7.92	95	.083368421			
D.u	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
^u LD.	.3501158	.084573	4.14	0.000	.1821466	.518085
^g --. L1.	-.1840843 -.0991552	.0306984 .0368244	-6.00 -2.69	0.000 0.008	-.245054 -.1722917	-.1231146 -.0260187
_cons	.3780104	.0578398	6.54	0.000	.2631356	.4928853

. estat bgodfrey

Breusch-Godfrey LM test for autocorrelation

lags(ρ)	chi2	df	Prob > chi2
1	0.170	1	0.6804

H0: no serial correlation

Model qiymatlarida avtokorrelatsiya qolmagan ko`rinadi (p -qiymat=0,68), bu ARDL (1,1) modelining adekvat ekanligini ko`rsatadi.

9.6.3. Avtoregressiv modellar

Avtoregressiv modellar ARDL (p,q) ning maxsus holatlari bo`lishi mumkin. Asosan, AR (p) modeli ARDL ($p,0$)ga ekvivalentdir. *okun.dta* da topilgan AQSH ning YAIM o`sishi haqidagi ma`lumotlar 9.3- bo`limda avtokorrelatsiya uchun tekshiriladi. g ning korrelogrammasida vaqt qatori kuzatuvlari o`rtasida korrelatsiya mavjudligini tasdiqlovchi dalillar mavjud edi.


Buni batafsil o`rganish uchun YAIM o`sishi uchun AR(2) modeli taxmin qilamiz va qoldiqlarning korrelogrammasi tuziladi. Dastlabki 12 kechikish uchun avtoregressiya, saqlangan qoldiqlar va avtokorrelatsiyalar taxmin qilinadi:

reg g L(1/2).g

predict ehat, res

ac ehat, lags(12)

Bu quyidagi hosil qiladi:



AR(2) model adekvat ko`rinadi, chunki qoldiq avtokorrelyatsiyalarning aksariyati kichik va ahamiyatsizdir. Buni batafsil o`rganish uchun avtoregressiya tartibi 1 dan 5 gacha o`zgarib turadi va model tanlash qoidalari afzal modelni tanlash uchun ishlataladi, yana model AIC yoki SC ning eng kichik qiymatini ishlab chiqaradi:

```
forvalues p=1/5 {  
    qui reg L(0/^p').g if date>tq(1986q2)  
    display "p=`p'  
    modelsel  
}
```

Bu quyidagini hosil qiladi:

```
p=1  
    aic = -1.0935183  
    sc = -1.0390538  
    obs = 93  
p=2  
    aic = -1.130582  
    sc = -1.0488852  
    obs = 93  
p=3  
    aic = -1.1242025  
    sc = -1.0152735  
    obs = 93  
p=4  
    aic = -1.1331587  
    sc = -.99699743  
    obs = 93  
p=5  
    aic = -1.1116622  
    sc = -.94826871  
    obs = 93
```

AIC ning eng kichik qiymatini ishlab chiqaruvchi model AR(4) dir, bunda AIC=-1,133 ga teng. Eng kichik SC ishlab chiqaruvchi model AR (2) dir, bunda SC=-1,049 ga

teng. Bu ikkita model tanlash qoidalari o`rtasidagi muhim farqni ko`rsatadi: SC mezoni regressorni qo`shish uchun katta jarima soladi va AIC ga qaraganda kichikroq modellarni tanlashga intiladi. Bu amaliyotchilar o`rtasida yaxshi ma`lum va tushuniladi.

9.7. Prognozlash

Ushbu bo`limda biz 3 xil modellar, AR modeli, ARDL modeli va eksponensial tekislash modelidan foydalangan holda prognozlashni ko`rib chiqamiz. Misollar odatda keljakda 3 davrgacha bo`lgan qisqa muddatli prognozlashga qaratiladi.

9.7.1. AR modeli bilan prognoz qilish

Aytaylik, bu 2009 yilning 3-choragi va 2009 yilning 3-choragigacha bo`lgan ma`lumotlardan foydalangan holda YaIM o`sishining AR(2) modelini taxmin qildik. Ushbu bo`limda keyingi davrlarni bashorat qilish uchun AR(2) modelidan foydalanish muhokama qilinadi va keyingi uch davrni prognoz qilish muhokama qilinadi hamda prognozning ishonch intervallari yaratiladi.

Uning noma`lum koeffitsiyentlari bo`yicha AR(2) modeli.

$$G_t = \delta + \theta_1 G_{t-1} + \theta_2 G_{t-2} + \nu_t$$

Oxirgi namunali kuzatishni G_T deb belgilab, vazifa G_{T+1} , G_{T+2} va G_{T+3} ni bashorat qilishadi. Mavjud namunadan keyingi kuzatuvning qiymati:

$$G_{T+1} = \delta + \theta_1 G_T + \theta_2 G_{T-1} + \nu_{T+1}$$

So`nggi 2 chorak uchun o`sish sur`atlari $G_T = G_{2009q3} = 0,8$ va $G_{T-2} = G_{2009q2} = -0,2$ bo`lib, bu parametrлarning taxminiy qiymatlari bilan $G_{T+1} = G_{2009q4}$ prognozlarini yaratish uchun foydalaniladi.

$$\begin{aligned}\hat{G}_{T+1} &= \hat{\delta} + \hat{\theta}_1 G_T + \hat{\theta}_2 G_{T-1} \\ &= 0,46573 + 0,37700 * 0,8 + 0,24624 * (-0,2) \\ &= 0,7181\end{aligned}$$

Model Stata da taxmin qilinganidan keyin uni hisoblash oson. AR(2) modelini baholang.

reg g L(1/2).

Keyin yordamida skalyar prognozini hisoblang:

scalar ghat1 = _b[_cons]+_b[L1.g]*g[98]+_b[L2.g]*g[97]

Stataning indekslash qobiliyati ma`lumotlar to`plamidagi G dagi oxirgi ikkita kuzatishni olish uchun ishlataladi. Ma`limotlar 2009q3 da tugaydigan 98 ta kuzatuvdan iborat bo`lgani uchun g[98] G_{2009q3} bo`yicha 98-kuzatuvga ishora qiladi. Xuddi shunday, g[97] 2009q2 dan G dagi kuzatuvga ishora qiladi.

Keyingi prognoz

$$\hat{G}_{T+2} = \hat{\delta} + \hat{\theta}_1 \hat{G}_{T+1} + \hat{\theta}_2 G_T$$

Yuqoridagi yordamida

*scalar ghat2 = _b[_cons]+_b[L1.g]*ghat1+ _b[L2.g]*g[98]*

\widehat{G}_{T+1} ni baholash uchun *ghat1* prognozidan foydalanilganiga e'tibor bering. G_T haqiqatda kuzatiladi va 98-kuzatuvdagi ma'lumotlar to`plamida joylashgan.

*scalar ghat2 = _b[_cons]+_b[L1.g]*ghat1+ _b[L2.g]*g[98]*

Nihoyat, oxirgi prognoz

$$\widehat{G}_{T+3} = \widehat{\delta} + \theta_1 \widehat{G}_{T+2} + \theta_2 \widehat{G}_{T+1}$$

Natijada:

*scalar ghat3 = _b[_cons]+_b[L1.g]*ghat2+ _b[L2.g]*ghat1*

Prognoz *ghat2* G_{T+2} ni baholash uchun ishlataladi va bu prognoz butunlay oldingi prognozlarga bog`liq.

Shu tarzda yaratilgan prognozlarning to`liq to`plami:

```
. scalar list ghat1 ghat2 ghat3
    ghat1 = .71807948
    ghat2 = .93343472
    ghat3 = .99445191
```

POE4 adabiyotida ko`rsatilganidek, prognoz xatosi farqlari:

$$\begin{aligned}\sigma_1^2 &= \text{var}(u_1) = \sigma_v^2 \\ \sigma_2^2 &= \text{var}(u_2) = \sigma_v^2(1+\theta_1^2) \\ \sigma_3^2 &= \text{var}(u_3) = \sigma_v^2((\theta_1^2+\theta_2^2)^2+\theta_1^2+1)\end{aligned}$$

AR(2) modelidagi baholarni almashtirish orqali baholanadi.

scalar var = e(rmse)^2

scalar se1 = sqrt(var)

scalar se2 = sqrt(var(1+(_b[L1.g])^2))*

scalar se3 = sqrt(var((_b[L1.g]^2+_b[L2.g])^2+1+_b[L1.g]^2))*

scalar list se1 se2 se3

```
. scalar list se1 se2 se3
    se1 = .55268751
    se2 = .59065984
    se3 = .62845236
```

95% prognozli ishonch oraliqlari odatiy tarzda tuzilgan. Ular prognoz markazida joylashgan va har ikki yo`nalishda taxminan 2 ta standart og`ishlarni kengaytiradi. Aniqrog`i, ular t-distribution 2,5% kritik qiymatidan foydalangan holda hisoblanadi va yuqorida hisoblangan prognoz standart xatolaridan foydalanadi.

*scalar f1L = ghat1 - invtail(e(df_r),.025)*se1*

*scalar f1U = ghat1 + invtail(e(df_r),.025)*se1*

*scalar f2L = ghat2 - invtail(e(df_r),.025)*se2*

*scalar f2U = ghat2 + invtail(e(df_r),.025)*se2*

*scalar f3L = ghat3 - invtail(e(df_r),.025)*se3*

*scalar f3U = ghat3 + invtail(e(df_r),.025)*se3*

scalar list f1L f1U f2L f2U f3L f3U

Statada aniq kritik qiymatni hisoblash *invtail* funktsiyasidan foydalanadi. Natijalar:

```
. scalar list f1L f1U f2L f2U f3L f3U
  f1L = -.37944839
  f1U = 1.8156073
  f2L = -.23949866
  f2U = 2.1063681
  f3L = -.25352994
  f3U = 2.2424338
```

9.7.2. Eksponensial tekislash

O`zgaruvchining kelajakdagi qiymatini uning tarixiga asoslangan holda bashorat qilish uchun ishlataladigan yana bir mashhur model eksponensial tekislashdir. AR modeli bilan prognoz qilish kabi, eksponensial tekislash yordamida bashorat qilish boshqa o`zgaruvchilardan ma`lumotdan foydalanmaydi.

Asosiy g`oya shundan iboratki, keyingi davr uchun prognoz joriy davr uchun prognozning o`rtacha og`irligi va joriy davrda real realizatsiya qilingan qiymatdir.

$$\hat{y}_{T+1} = \alpha y_T + (1-\alpha) \hat{y}_T$$

Ekponentsial tekislash usuli ko`p qirrali prognozlash vositasidir, ammo \hat{y}_{T+1} ni yaratish uchun tekislash parametri α va \hat{y}_T uchun qiymat kerak bo`ladi. α qiymati joriy ma`lumotlarning nisbiy og`irligi haqidagi fikrni aks ettirishi mumkin; muqobil ravishda *within-sample forecasts* olish orqali tarixiy ma`lumotlardan taxmin qilish mumkin.

$$\hat{y}_t = \alpha y_{t-1} + (1-\alpha) \hat{y}_{t-1} \quad t=2,3,\dots,T$$

va kvadratlar yig`indisini minimallashtiradigan α qiymatini tanlash *one-step forecast errors*.

$$v_t = y_t - \hat{y}_t = y_t - (\alpha y_{t-1} + (1-\alpha) \hat{y}_{t-1})$$

α ning kichikroq qiymatlari prognozning yanada tekislanishiga olib keladi. Stata dasturida *tssmooth* vaqtli qatorlar uchun tekislashning turli shakllarini amalga oshiradigan tartib mavjud. *tssmooth* yangi o`zgaruvchini *newvar* ko`rinishida yaratadi va o`zgaruvchini so`ralgan sillqlash vositasidan o`tkazish orqali uni to`ldiradi. Eksponensialni o`z ichiga olgan bir nechta sillqlash vositalari mavjud. Bir marta tekislash parametrining kerakli qiymatini belgilashi mumkin, α yoki uning qiymati POE4 adabiyotda muhokama qilingandek, namunadagi kvadratlarning yig`indisini bashorat qilish xatolarini minimallashtirish uchun avtomatik ravishda tanlanishi mumkin.

Quyida, *okun.dta* ma`lumotlari YaIM o`sishining eksponensial tekislangan prognoz qiymatlarini olish uchun ishlataladi. Avval ma`lumotlar ochiladi, yaratilgan sanalar, qayta formatlanadi va o`zgaruvchilar vaqtli qatorlar sifatida o`rnatiladi.

use okun, clear

generate date = tq(1985q2) + _n-1

format %tq date

tsset date

Tekislashdan **g** oldin qilinadigan birinchi narsa vaqt qatoriga (bo`sh) kuzatuvni qo`shishdir. Buni tekislashdan oldin qilish Stataga ushbu kuzatuvni bir qadam oldinga bashorat bilan to`ldirishga imkon beradi.

tsappend, add(1)

g qatorni eksponent tekislash buyrug`i:

tssmooth exponential sm1=g, parms(.38)

Quyida keltirilgan **tssmooth** sintaksisi biroz tushintirishga loyiqidir.

Syntax

tssmooth smoother [type] newvar = exp [if] [in] [, ...]

Smoother category	smoother
Moving average with uniform weights with specified weights	<u>ma</u>
Recursive exponential double exponential nonseasonal Holt-Winters seasonal Holt-Winters	<u>exponential</u> <u>dexponential</u> <u>hwinters</u> <u>swinters</u>
Nonlinear filter	<u>n1</u>

Ko`rsatish kerak bo`lgan birinchi narsalar-kerakli tekislash turi hisoblanadi. Bu yerda biz eksponensialni tanlaymiz. Keyin yangi o`zgaruvchi nomi yaratilishi va tekislanadigan qatorga teng bo`lishi kerak (**sm1=g**). Bundan keyin ba`zi variantlar keladi. Birinchi, **parm(.38)**, tekislash parametrining qiymatini o`rnatadi. Agar bu parametr ko`rsatilmagan bo`lsa, u holda **tssmooth** kvadrat xatolar yig`indisini minimallashtiradiganini tanlaydi.


Bu quyidagi natijalarni chiqaradi:

. **tssmooth exponential sm1=g, parms(.38)**

exponential coefficient =	0.3800
sum-of-squared residuals =	31.122
root mean squared error =	.56354

Yangi o`zgaruvchi **sm1** eksponent tekislangan qatorni o`z ichiga oladi va u ma`lumotlar to`plamiga qo`shiladi. Tekislangan qator hosil bo`lgach, uni vaqtli qatorlardagi tekislanmagan versiya bilan solishtirish mumkin. Keyingi satrda ikkita seriya chiziladi va rivoyatning nomi o`zgartiriladi, shunda hamma narsa grafikaga biroz yaxshiroq mos tushadi.

tsline sm1 g, legend/lab(1 "G") lab(2 "Ghat") title(alpha=.38) ||| lpattern(solid dash)



Keyingi davr uchun prognoz tuziladi va ro`yhatga olinadi. Shuningdek, ma`lumot **tssmooth** tomonidan 1 davr uchun avtomatik ravishda hisoblanadi. Agar ko`proq so`ralsa, Stata buning uchun variantlarni taklif qiladi. Qo`lda yaratilgan va Stata dan avtomatik prognoz mos keladi.

```
scalar f1 = .38*g[98]+(1-.38)*sm1[98]
```

```
scalar list f1
```

```
list sm1 in 99
```

```
. scalar list f1  
f1 = .05356533  
. list sm1 in 99
```

	sm1
99.	.0535653

$\alpha = 0.8$ tekislash parametri uchun mashq takrorlanadi. Buyruq va natijalar quyida keltirilgan.


```
tssmooth exponential sm2=g, parms(.8)
```

```
tsline sm2 g, legend/lab (1 "G") lab(2 "Ghat") title(alpha=.8) |||
```

```
lpattern(solid dash)
```

```
scalar f2 = .8*g[98]+(1-.8)*sm2[98]
```

```
scalar list f2
```



$\alpha = 0,8$ ning kattaroq qiymati kamroq silliqlikka olib keladi; eksponensial tekislangan qator $\alpha = 0,38$ bo`lganda asl nusxaga ancha yaqinroq. Prognoz, $f_2, 0,56128444$ bu kattaroq tekislash parametri bilan yaratilganidan ancha katta.

Nihoyat, agar yumshatuvchi parametrning o`ziga xos qiymati variant sifatida o`tkazib yuborilsa, Stata namunadagi kvadratlar yig`indisini bashorat qilish xatolarini minimallashtiradigan qiymatni tanlaydi. Bu Stata yordamida baholanadi

tssmooth exponential sm3=g

scalar f3 = r(alpha)*g[98]+(1-r(alpha))*sm3[98]

scalar list f3

Natija:

```
computing optimal exponential coefficient (0,1)

optimal exponential coefficient =      0.3803
sum-of-squared residuals       =    31.122043
root mean squared error        =   .56353515

. scalar f3 = r(alpha)*g[98]+(1-r(alpha))*sm3[98]
. scalar list f3
f3 = .05367152
```

Bu hisob-kitoblarga ko`ra, 0,38 ning belgilangan tanlovi ma`lumotli edi! Tekislash parametrining qiymati tekislashdan keyin **r(alpha)** sifatida saqlanadi va u prognozlarni xuddi belgilangan qiymat bilan osonlik bilan yaratish uchun ishlatilishi mumkinligini unutmang.

9.8. Multiplikator tahlili

Multiplikator tahlili bir o`zgaruvchidagi o`zgarishning boshqa o`zgaruvchining natijasiga ta`siri va ta`sir qilish vaqtini bildiradi. Multiplikator tahlilining eng oddiy shakli cheklangan taqsimlangan kechikish modeliga asoslanadi.

$$y_t = \alpha + \beta_0 x_t + \beta_1 x_{t-1} + \beta_2 x_{t-2} + \dots + \beta_q x_{t-q} + e_t$$

Ushbu modeldan hisoblangan koeffitsientlar *impact*, *delay* va *interim multiplikatorlarni* ishlab chiqarish uchun ishlatilishi mumkin. Ta`sir ko`paytmasi- x_t ning bir birlikka o`zgarishining y_t o`rtacha qiymatiga ta`siri. x va y bir xil vaqt oralig`ida bo`lgani uchun ta`sir bir vaqtida va shuning uchun o`zgarishning dastlabki ta`siriga teng. S-davr *delay multiplier* hisoblanadi.

$$\frac{\partial E(y_t)}{\partial x_{t-s}} = \beta_s$$

O`tmishdagi x s-davrlardagi o`zgarishning joriy davrdagi qaram o`zgaruvchining o`rtacha qiymatiga ta`siri. Agar x_t 1 birlikka oshirilsa va keyingi davrlarda $(t+1), (t+2), \dots$, yangi darajasida saqlansa, oraliq multiplikatorni hisoblash mumkin. *Interim multiplier* to`plangan effektni o`lchash uchun darhol ta`sirni (ta`sir multiplikatori) β_0 ni keyingi kechikish ko`paytirgichlariga qo`shadi. Shuning uchun $t+1$ davrida oraliq effect $\beta_0 + \beta_1$ ga teng. $t+2$ davrida $\beta_0 + \beta_1 + \beta_2$ bo`ladi va hokaza. *total multiplicator* q yoki undan ortiq davrlar o`tgandan keyin barqaror o`sishning y ga yakuniy ta`siri; u $\sum_{s=0}^q \beta_s$ bilan berilgan.

ARDL modeli AR modeliga qaram o`zgaruvchining kechikkan qiymatlarini qo`shadi.

$$y_t = \delta + \theta_1 y_{t-1} + \dots + \theta_p y_{t-p} + \delta_0 x_t + \delta_1 x_{t-1} + \dots + \delta_q x_{t-q} + \nu_t$$

va bu multiplikator tahlilini biroz qiyinlashtiradi. Asosan, bu kechikish operatori L xususiyatlaridan foydalangan holda cheksiz taqsimlangan kechikish modeliga aylantirilishi kerak, bu xuddi unga asoslangan Stata buyruqlari kabi ishlaydi. Ya`ni $L^i x_t = x_{t-i}$. Bu modelni tanish AR shakliga kiritadi va multiplikatorlarning odatiy ta`riflari qo`llanilishi mumkin. Bu POE4 adabiyotida batafsil muhokama qilinadi va bu yerda takrorlanmaydi.

Okun qonunini tasvirlash uchun foydalilanligan ARDL (1,1) modeli uchun bizda mavjud.

$$DU_t = \delta + \theta_1 DU_{t-1} + \delta_0 G_t + \delta_1 G_{t-1} + \nu_t$$

yoki kechikish operatori bilan L yozilgan

$$(1 - \theta_1 L) DU_t = \delta + (\delta_0 + \delta_1 L) G_t + \nu_t$$

$$DU_t = (1 - \theta_1 L)^{-1} + \delta + (1 - \theta_1 L)^{-1} (\delta_0 + \delta_1 L) G_t + (1 - \theta_1 L)^{-1} \nu_t$$

$$\begin{aligned} DU_t &= \alpha + \beta_0 G_t + \beta_1 G_{t-1} + \beta_2 G_{t-2} + \beta_3 G_{t-3} + \dots + e_t \\ &= \alpha + (\beta_0 + \beta_1 L + \beta_2 L^2 + \beta_3 L^3 + \dots) G_t + e_t \end{aligned}$$

Bu shunchaki cheksiz taqsimlangan kechikish modeli. Multiplikatorlar uchun koeffitsientlar β_s ni o`z ichiga oladi, ular ARDL ning taxminiy parametrlari nuqtai nazaridan hal qilinishi kerak.

POE4 adabiyotida berilgan yechimlar:

$$\beta_0 = \delta_0$$

$$\beta_1 = \delta_1 + \beta_0 \theta_1$$

$$\beta_j = \beta_{j-1} \theta_1 \quad \text{bu uchun} \quad j \geq 2$$

Okun modeli uchun ARDL (1,1) ga asoslangan ta`sirini va dastlabki bir necha oraliq ko`paytirgichlarni baholash uchun Stata buyrug`i:

regress D.u L.D.u L(0/1).g

scalar b0 = _b[g]

scalar b1 = _b[L1.D.u]*b0+_b[L1.g]

scalar b2 = b1*_b[L1.D.u]

scalar b3 = b2*_b[L1.D.u]

va boshqalar.

Stata ularni grafik qilish uchun ma`lumotlar to`plamiga kiritishning silliq usulini taqdim etadi. Regressiyadan so`ng **mult** deb nomlangan yangi o`zgaruvchini yarating va β_0 hisoblangan koeffitsiyentini birinchi kuzatuvga qo`ying.

gen mult = _b[g] in 1

Ikkinchi kuzatuv uchun $\beta_1 = \delta_1 + \beta_0\theta_1$ hisoblangan qiymatni ikkinchi kuzatishga kiritish uchun almashtirish buyrug`idan foydalaning:

replace mult = L.mult*_b[L1.D.u]+_b[L1.g] in 2

β_0 ni baholash uchun **L.mult** ishlatalganiga e`tibor bering. Qolgan ko`paytirgichlar $\beta_j = \beta_{j-1}\theta_1$ asosida hisoblanadi, uni bitta chiziq yordamida hisoblash mumkin.

replace mult = L.mult*_b[L1.D.u] in 3/8

list mult in 1/8

Bu holda **L.mult** ko`paytirgichlarni o`z ichiga olgan **mult** ning kechikish qiymati. Agar xoxlasangiz, ularni T gacha osongina hisoblappingiz mumkin, biz faqat sakkiztasini tanladik. Nihoyat, kechikish og`irliliklari (1dan 8 gacha) sifatida ishlataladigan butun sonlarni o`z ichiga olgan **lag** deb nomlangan yangi o`zgaruvchini yarating. Nihoyat, siz ularni chizishingiz mumkin.

gen lag = _n-1 in 1/8


line mult lag in 1/8

Ko`paytiruvchilar quyidagilardir:

. list mult in 1/8

mult	
1.	- .1840843
2.	- .163606
3.	- .057281
4.	- .020055
5.	- .0070216
6.	- .0024584
7.	- .0008607
8.	- .0003013

va grafikda:



Dastlabki ta'sir salbiy ko`rinadi, lekin vaqt o'tishi bilan nolga yaqinlashadi. 6-davrga kelib YaIM o'sishining bir birlik o'zgarishining ishsizlikka ta'siri deyarli nolga teng.

9.9. QO'SHIMCHA TESTLAR

9.9.1. Durbin-Watson testi

Durbin -Watson statistikasi regressiyadan keyin *estat dwatson* yordamida osongina natijalarini chiqaramiz. Phillips ma'lumotlar to`plami uchun DW statistikasi quyidagi buyruq yordamida topiladi:

```
* Durbin Watson test
use phillips_aus, clear
generate date = tq(1987q1) + _n-1
format %tq date
tsset date
```

```
regress inf D.u
estat dwatson
```

Natija esa:

```
. estat dwatson
Durbin-Watson d-statistic( 2,    90) = .8872891
```

E'tibor bering, Statadagi *dwatson* testi jadvaldagи yuqori va pastki chegaralarni ko'rib chiqishni talab qiladi. DW ning taqsimlash funksiyasini integratsiyalash natijasida olingan aniq p -qiymati hozirgi vaqtida bajarilmaydi.

9.9.2. Prais-Winsten FGLS

AR(1) modelining amalga oshirilishi mumkin bo`lgan GLS baholovchisi *prais* deb nomlangan Stata protsedurasi yordamida baholanishi mumkin. *Prais* buyrug'i *regress* kabi ishlaydi va shunga o`xshash sintaksisdan foydalanadi. Agar qiziqsangiz, o`rganishga arziydigan bir nechta qo'shimcha variantlar mavjud. *Prais* ning eng katta cheklovi shundaki, u faqat birinchi darajali avtokorrelyastsiyaga ega modellarni baholaydi. Murakkabroq

modellar uchun maksimal ehtimollik yordamida umumiyroq modellarni baholaydigan **arima** buyrug`iga qarang. AR(1) xatoliklari bilan chiziqli regressiya modelining ikki bosqichli FGLS bahosi taxmin qilingan.

* **Prais-Winsten FGLS estimator**

prais inf D.u, twostep

Agar **twostep** variant berilmasa, hisoblagich barqaror yechim topilguncha takrorlanadi. Ikkala baholovchi ham bir xil asimptotik xususiyarlarga ega, shuning uchun takrorlashning hojati yo`q.

prais inf D.u

bu quyidagi natijani hosil qiladi:

> esttab _2step Iterate, compress se(%12.3f) b(%12.5f) gaps scalars(rss rho) //		
> /		
mtitle("2-step" "Iterated") title("Dependent variable: inf")		
Dependent variable: inf		
(1)	(2)	
2-step	Iterated	
D.u	-0.69943** (0.243)	-0.70236** (0.243)
_cons	0.78584*** (0.120)	0.78619*** (0.122)
N	90	90
rss	23.50157	23.49538
rho	0.54988	0.55825
Standard errors in parentheses		
* p<0.05, ** p<0.01, *** p<0.001		

Birinchi ustun ikki bosqichli FGLS natijalarini, ikkinchisi esa iteratsiya natijalarini o`z ichiga oladi. Natijalar juda o`xshash.

Arima buyrug`i orqali olingan maksimal ehtimollik taxminchisi yordamida baholanadi.

arima inf D.u, ar(1)

Natija:

ARIMA regression						
Sample: 1987q2 - 2009q3				Number of obs	=	90
Log likelihood = -67.4559				Wald chi2(2)	=	44.96
inf	Coef.	OPG Std. Err.	z	P> z	[95% Conf. Interval]	
inf						
<u>D1.</u>	-.7025681	.3167053	-2.22	0.027	-1.323299	-.0818371
_cons	.7861493	.1398032	5.62	0.000	.51214	1.060159
ARMA						
_ar_L1.	.5588218	.0873961	6.39	0.000	.3875285	.7301151
/sigma	.5109273	.0277513	18.41	0.000	.4565358	.5653188

Birinchi blok regressiya taxminlarini o`z ichiga oladi. Kesishish 0,786 ishsizlikning o`zgarishi bo`yicha qiyalik – 0,7023 deb baholanmoqda. Avtokorrelyatsiya parametri

ARMA etiketli qutida 0,559 deb baholanadi. Bu natijalar **prais** yordamida olingan FGLS baholariga juda o`xshash.

IX bob mavzularini mustahkamlash uchun savollar

1. Vaqtli qatorlar ma'lumotlarini tahlil qilish deganda nimani tushunasiz?
2. Psevdofunktsiya deb nimaga aytildi?
3. Vaqtli qatorlar bilan ishslashda **format** buyrug'i qanday vazifani bajaradi?
4. **tsset** buyrug'i vaqtli qatorlarda qaysi buyruqdan so'ng amalga oshiriladi?
5. *okun.dta* ma'lumotlari nima uchun xizmat qiladi?
6. Vaqtli qatorlar o`zgaruvchilaridan foydalanishni osonlashtirish uchun xizmat qiluvchi buyruq qaysi?
 7. Okun qonu qonuni mazmuni qanday?
 8. Ishsizlik darajasi formulasi qanday ifodalanadi?
 9. **return list** buyrug'i nima mazifasi bajaradi?
 10. Fillips egri chizig'i nimani ifodalaydi?
 11. O`rtacha og`irlilik qanday buyruq asosida amalga oshiriladi?
 12. Qaysi buyruq orqali ko`rib chiqilayotgan modellarning turli chiziqli xususiyatlari taqqoslanadi?
 13. Ekponentsiyal tekislash usuli nima?
 14. Multiplikator tahlili nimani anglatadi?